4.12 Tukey's Test for Nonadditivity

- Consider a two-factor $a \times b$ factorial design that has only $n = 1$ replicate for each of the ab treatment combinations. We sometimes refer to this as an **unreplicated** experiment or an experiment with a **one observation per cell design**.
- Recall that the MSE df $= ab(n-1)$ for the two-factor interaction model. Thus, there will be 0 df for the MSE when $n = 1$. That is, the df column in the ANOVA table looks like:

Source of Variation	*df*
A	$a-1$
B	$b-1$
AB	$(a-1)(b-1)$
Error	0
Total	$ab-1$

- If no interaction exists (i.e., we assume all $(\alpha\beta)_{ij}$ effects = 0), then the additive model $y_{ij} = \mu + \alpha_i + \beta_j + \epsilon_{ij}$ is assumed to be adequate.
- Then the $(a-1)(b-1)$ df, SS, and MS for the interaction effects will now be the SS, df, and MS for the Error. This will allow us to perform tests on the main effects for A and B.
- If an interaction does exist, then what can we do? If we can assume the interaction effects possess a certain structure, then tests may exist.
- One such test is **Tukey's Test for Additivity** which tests if the interaction terms take on the form $(\alpha\beta)_{ij} = \gamma\tau_i\beta_j$. This says the $(ij)^{th}$ interaction effect is proportional to the product of the main effects.
- If this interaction structure is true, then the model will be:

$$y_{ij} =$$

- Note, that we have added only 1 parameter (γ) to the additive model leaving us $(a-1)(b-1)-1 = ab - a - b$ df for the new MSE.
- The following SAS code tricks PROC GLM into fitting this model and performing Tukey's Test for Nonadditivity.

SAS Code for Tukey's Test for Additivity Example

```
*********************************************************;
*** TUKEY'S NONADDITIVITY TEST FOR A TWO-FACTOR DESIGN ***;
***           WITH ONE OBSERVATION PER CELL           ***;
*********************************************************;

DATA in; INPUT temp pressure impurity @@; CARDS;
100 25 5   100 30 4   100 35 6   100 40 3   100 45 5
125 25 3   125 30 1   125 35 4   125 40 2   125 45 3
150 25 1   150 30 1   150 35 3   150 40 1   150 45 2
 ;
PROC GLM DATA=in NOPRINT;
     CLASS temp pressure;
     MODEL impurity = temp pressure ;
     OUTPUT OUT=diag PREDICTED=yhat;

DATA diag; SET diag;
     nonadd = yhat**2;

TITLE 'TUKEY TEST FOR NONADDITIVITY -- 1 OBS PER CELL';

PROC GLM DATA=diag;
     CLASS temp pressure;
     MODEL impurity = temp pressure nonadd / SS1 SOLUTION;
RUN;
```

TUKEY TEST FOR NONADDITIVITY -- 1 OBS PER CELL

The GLM Procedure

Variable: IMPURITY

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	7	35.03185550	5.00455079	18.42	0.0005
Error	7	1.90147783	0.27163969		
Corrected Total	14	36.93333333			

R-Square	Coeff Var	Root MSE	IMPURITY Mean
0.948516	17.76786	0.521191	2.933333

Source	DF	Type I SS	Mean Square	F Value	Pr > F
TEMP	2	23.33333333	11.66666667	42.95	0.0001
PRESSURE	4	11.60000000	2.90000000	10.68	0.0042
NONADD	1	0.09852217	0.09852217	0.36	0.5660

Parameter	Estimate		Standard Error	t Value	Pr > |t|
Intercept	1.812807882	B	0.47262874	3.84	0.0064
TEMP 100	2.312807882	B	1.18771619	1.95	0.0925
TEMP 125	0.844827586	B	0.41838222	2.02	0.0832
TEMP 150	0.000000000	B	.	.	.
PRESSURE 25	-0.255336617	B	0.44482151	-0.57	0.5839
PRESSURE 30	-1.070607553	B	0.60942962	-1.76	0.1224
PRESSURE 35	0.716748768	B	0.63427295	1.13	0.2957
PRESSURE 40	-1.070607553	B	0.60942962	-1.76	0.1224
PRESSURE 45	0.000000000	B	.	.	.
NONADD	0.036945813		0.06134722	0.60	0.5660

4.13 The Two-Factor Random Effects Model

- Suppose both factors in the two-factor factorial design are random. Thus, the a levels of factor A and the b levels of factor B are randomly selected from a large population of levels.

- For this experimental situation, the random effects model is:

$$y_{ijk} \;= \tag{25}$$

where τ_i, β_j, $(\tau\beta)_{ij}$ and ϵ_{ijk} are **random effects**.

- Thus, there will be variance components σ^2_τ, σ^2_β, and $\sigma^2_{\tau\beta}$ associated with these random effects.

- To test hypotheses regarding the model, we assume:

$\tau_i \sim$ IID $N(0, \sigma^2_\tau)$ $\qquad$ $\beta_j \sim$ IID $N(0, \sigma^2_\beta)$.
$(\tau\beta)_{ij} \sim$ IID $N(0, \sigma^2_{\tau\beta})$ $\qquad$ $\epsilon_{ijk} \sim$ IID $N(0, \sigma^2)$.
where τ_i, β_j, $(\tau\beta)_{ij}$ and ϵ_{ijk} are independent for all i, j, k.

- A hypothesis test about the equality of means or effects is not correct in the random effects case. The hypotheses of interest involve the three variance components:

$$H_0:\ \sigma^2_\tau = 0 \qquad \text{and} \qquad H_1:\ \sigma^2_\tau > 0 \tag{26}$$

$$H_0:\ \sigma^2_\beta = 0 \qquad \text{and} \qquad H_1:\ \sigma^2_\beta > 0 \tag{27}$$

$$H_0:\ \sigma^2_{\tau\beta} = 0 \qquad \text{and} \qquad H_1:\ \sigma^2_{\tau\beta} > 0 \tag{28}$$

- To test the hypotheses in (26), (27), and (28), we use the same mean squares from the fixed effects ANOVA, but the <u>F-statistics are different</u>. To determine the appropriate F-statistics, we need to examine the expected mean squares for the random effects model in (25).

4.13.1 Expected Mean Squares and Variance Component Estimation

- For equal cell sizes (all $n_{ij} = n$):

$$\begin{aligned} E(MS_A) &= \sigma^2 + n\sigma^2_{\tau\beta} + bn\sigma^2_\tau \\ E(MS_B) &= \sigma^2 + n\sigma^2_{\tau\beta} + an\sigma^2_\beta \\ E(MS_{AB}) &= \sigma^2 + n\sigma^2_{\tau\beta} \\ E(MS_E) &= \sigma^2 \end{aligned} \tag{29}$$

- If H_0 is true in (26), then $\sigma^2_\tau = 0$, and $E(MS_A) =$ which matches

- If H_0 is true in (27), then $\sigma^2_\beta = 0$, and $E(MS_B) =$ which matches

  This implies $F_A = \dfrac{MS_A}{MS_{AB}}$ and $F_B = \dfrac{MS_B}{MS_{AB}}$ are the F-statistics for the tests in (26) and (27).

- If H_0 is true in (28), then $\sigma^2_{\tau\beta} = 0$, and $E(MS_{AB}) =$ which matches

  This implies $F_{AB} = \dfrac{MS_{AB}}{MS_E}$ is the F-statistic used to test H_0 in (28).

- In summary, the ANOVA test statistics for the two-factor random effects model are

$$F_A = \frac{MS_A}{MS_{AB}} \text{ for (26)}, \qquad F_B = \frac{MS_B}{MS_{AB}} \text{ for (27)}, \qquad F_{AB} = \frac{MS_{AB}}{MS_E} \text{ for (28)}$$

- For the random effects model, replace $E(MS_A)$, $E(MS_B)$, $E(MS_{AB})$, and $E(MS_E)$ in (29) with the calculated values of MS_A, MS_B, MS_{AB}, and MS_E.
- Solving the system of equations produces estimates of the variance components:

$$\widehat{\sigma}^2 = MS_E \qquad \widehat{\sigma}^2_{\tau\beta} = \frac{MS_{AB} - MS_E}{n}.$$

$$\widehat{\sigma}^2_{\tau} = \frac{MS_A - MS_{AB}}{bn}. \qquad \widehat{\sigma}^2_{\beta} = \frac{MS_B - MS_{AB}}{an}.$$

- For unequal sample sizes, the coefficients of the variance components in the EMS is a complicated function of the n_{ij}. *SAS* will generate them using the RANDOM statement.

4.13.2 Random Effects Model Example

During a random workday, 4 production runs were randomly selected from the large number of daily production runs. Six batches of raw material were randomly selected from the large number of available batches. In each of the 4 runs, the same 6 batches of raw materials were used to make a synthetic fiber. Three random sample were taken from each (production run)*batch combination. The $N = 72$ samples were randomized before taking strength measurements.

	Production Run			
Batch	1	2	3	4
1	15.16 16.41 15.47	19.07 17.44 18.16	23.22 22.51 24.67	23.10 22.71 22.94
2	24.20 23.99 23.40	25.75 25.12 25.08	32.40 31.52 31.93	30.58 30.26 31.02
3	21.21 21.32 19.95	25.77 26.15 26.15	27.89 27.89 28.28	29.13 28.50 28.68
4	12.24 13.18 12.83	15.07 15.14 15.07	21.58 21.11 20.96	22.14 21.72 22.11
5	22.83 22.75 22.46	26.23 25.11 25.81	30.54 30.67 30.34	33.39 33.63 32.69
6	23.61 24.82 24.84	25.24 25.22 25.02	33.04 31.98 31.76	32.93 33.85 34.22

- By default, SAS assumes all effects are fixed. To get the EMS and the correct F-tests we use the RANDOM statement.
- For random factors A and B, the RANDOM statement has the form **RANDOM A|B / TEST**.
- To get estimates of the variance components, use PROC VARCOMP. An example is now given.
- The estimates of the variance components are

$$\widehat{\sigma}^2_{\tau} = \qquad \widehat{\sigma}^2_{\beta} = \qquad \widehat{\sigma}^2_{\tau\beta} = \qquad \widehat{\sigma}^2 =$$

- This indicates the largest sources of variability are due to variability across production runs and to variability across batches of raw material.
- In the EMS output in SAS, replace 'Var' with 'σ^2'. This will match the notation in the course notes. For this example, the EMS output in SAS is

Source	Type III Expected Mean Square
prodrun	Var(Error) + 3 Var(prodrun*batch) + 18 Var(prodrun)
batch	Var(Error) + 3 Var(prodrun*batch) + 12 Var(batch)
prodrun*batch	Var(Error) + 3 Var(prodrun*batch)

These are the same as

$$\begin{array}{l} \sigma^2 \ + \ 3\sigma^2_{\tau\beta} \ + \ 18\sigma^2_{\tau} \\ \sigma^2 \ + \ 3\sigma^2_{\tau\beta} \ + \ 12\sigma^2_{\beta} \\ \sigma^2 \ + \ 3\sigma^2_{\tau\beta} \end{array}$$

TWO-FACTOR RANDOM EFFECTS MODEL

The GLM Procedure

Dependent Variable: strength

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	23	2295.503778	99.804512	385.75	<.0001
Error	48	12.418933	0.258728		
Corrected Total	71	2307.922711			

R-Square	Coeff Var	Root MSE	strength Mean
0.994619	2.053826	0.508653	24.76611

Source	DF	Type III SS	Mean Square	F Value	Pr > F
prodrun	3	920.650922	306.883641	1186.13	<.0001
batch	5	1320.720844	264.144169	1020.93	<.0001
prodrun*batch	15	54.132011	3.608801	13.95	<.0001







Level of prodrun	N	strength Mean	strength Std Dev
1	18	20.0372222	4.50222665
2	18	22.5888889	4.43913066
3	18	27.9050000	4.36495839
4	18	28.5333333	4.75136141

Level of batch	N	strength Mean	strength Std Dev
1	12	20.0716667	3.45953448
2	12	27.9375000	3.58781049
3	12	25.9100000	3.26164487
4	12	17.7625000	4.12235177
5	12	28.0375000	4.29638775
6	12	28.8775000	4.33748801

Source	Type III Expected Mean Square
prodrun	Var(Error) + 3 Var(prodrun*batch) + 18 Var(prodrun)
batch	Var(Error) + 3 Var(prodrun*batch) + 12 Var(batch)
prodrun*batch	Var(Error) + 3 Var(prodrun*batch)

The GLM Procedure
Tests of Hypotheses for Random Model Analysis of Variance

Variable: strength

Source	DF	Type III SS	Mean Square	F Value	Pr > F
prodrun	3	920.650922	306.883641	85.04	<.0001
batch	5	1320.720844	264.144169	73.19	<.0001
Error	15	54.132011	3.608801		
Error: MS(prodrun*batch)					

Source	DF	Type III SS	Mean Square	F Value	Pr > F
prodrun*batch	15	54.132011	3.608801	13.95	<.0001
Error: MS(Error)	48	12.418933	0.258728		

TWO-FACTOR RANDOM EFFECTS MODEL VARIANCE COMPONENTS ANALYSIS

Variance Components Estimation Procedure

Dependent Variable:	strength

REML Iterations					
Iteration	**Objective**	**Var(prodrun)**	**Var(batch)**	**Var(prodrun*batch)**	**Var(Error)**
0	-0.5824695182	16.8486022222	21.7112806790	1.1166909877	0.2587277778
1	-0.5824695182	16.8486022222	21.7112806790	1.1166909877	0.2587277778

Convergence criteria met.

REML Estimates	
Variance Component	**Estimate**
Var(prodrun)	16.84860
Var(batch)	21.71128
Var(prodrun*batch)	1.11669
Var(Error)	0.25873

SAS Code for Random and Mixed Effects Model Examples

```
DM 'LOG; CLEAR; OUT; CLEAR;';

ODS GRAPHICS ON;
ODS PRINTER PDF file='C:\COURSES\ST541\RANMIX.PDF';
OPTIONS NODATE NONUMBER;
DATA in;
     DO prodrun = 1 to 4;
     DO batch = 1 to 6;
     DO rep = 1 to 3;
        INPUT strength @@; OUTPUT;
     END; END; END;
CARDS;
15.16 16.41 15.47 24.20 23.99 23.40 21.21 21.32 19.95
12.24 13.18 12.83 22.83 22.75 22.46 23.61 24.82 24.84
19.07 17.44 18.16 25.75 25.12 25.08 25.77 26.15 26.15
15.07 15.14 15.07 26.23 25.11 25.81 25.24 25.22 25.02
23.22 22.51 24.67 32.40 31.52 31.93 27.89 27.89 28.28
21.58 21.11 20.96 30.54 30.67 30.34 33.04 31.98 31.76
23.10 22.71 22.94 30.58 30.26 31.02 29.13 28.50 28.68
22.14 21.72 22.11 33.39 33.63 32.69 32.93 33.85 34.22
;
******************************************;
*** TWO-FACTOR RANDOM EFFECTS MODEL   ***;
*** WITH VARIANCE COMPONENTS ANALYSIS ***;
******************************************;

PROC GLM DATA=in;
     CLASS prodrun batch;
     MODEL strength = prodrun|batch / SS3;
     MEANS prodrun batch;
     RANDOM prodrun|batch / TEST;
TITLE 'TWO-FACTOR RANDOM EFFECTS MODEL';

PROC VARCOMP DATA=in METHOD=REML;
     CLASS prodrun batch;
     MODEL strength = prodrun|batch;
TITLE2 'VARIANCE COMPONENTS ANALYSIS';

******************************************;
*** TWO-FACTOR MIXED EFFECTS MODEL    ***;
*** WITH VARIANCE COMPONENTS ANALYSIS ***;
******************************************;

PROC GLM DATA=in;
     CLASS prodrun batch;
     MODEL strength = batch|prodrun / SS3;
     RANDOM batch prodrun*batch / TEST;
TITLE 'TWO FACTOR MIXED EFFECTS MODEL';

PROC VARCOMP DATA=in METHOD=REML;
     CLASS prodrun batch;
     MODEL strength = prodrun batch prodrun*batch / FIXED=1;
TITLE2 'VARIANCE COMPONENTS ANALYSIS';
RUN;
```

4.14 The Two-Factor Mixed Effects Model

- In a two-factor factorial design assume factor A is fixed while factor B is random. The a levels of factor A are fixed by the experimenter while the b levels of factor B are randomly selected from a large population of levels.
- For this experimental situation, the random effects model is:

$$y_{ijk} \; = \; \mu \; + \; \tau_i \; + \; \beta_j \; + \; (\tau\beta)_{ij} \; + \; \epsilon_{ijk} \tag{30}$$

  where β_j, $(\tau\beta)_{ij}$ and ϵ_{ijk} are random effects.
- Thus, there will be variance components σ^2_β and $\sigma^2_{\tau\beta}$ associated with these random effects.
- To test hypotheses regarding the model, we assume:

  $\beta_j \sim$ IID $N(0, \sigma^2_\beta)$ $\qquad$ $(\tau\beta)_{ij} \sim$ IID $N(0, \sigma^2_{\tau\beta})$ $\qquad$ $\epsilon_{ijk} \sim$ IID $N(0, \sigma^2)$
  where β_j, $(\tau\beta)_{ij}$ and ϵ_{ijk} are independent for all i, j, k.
- The hypotheses of interest involve (i) equality of effects for the fixed effect and (ii) if the variance components are 0:

$$H_0 : \tau_1 = \tau_2 = \cdots = \tau_a \qquad \text{and} \qquad H_1 : \tau_i \neq \tau_j \;\; for \; some \; i \neq j \tag{31}$$

$$H_0 : \; \sigma^2_\beta = 0 \quad \text{and} \quad H_1 : \; \sigma^2_\beta > 0 \tag{32}$$

$$H_0 : \; \sigma^2_{\tau\beta} = 0 \quad \text{and} \quad H_1 : \; \sigma^2_{\tau\beta} > 0 \tag{33}$$

- To test these we use the same mean squares from the fixed effects ANOVA but the F-statistics differ. To determine the appropriate F-statistics, we need to look at the expected mean squares for the mixed effects model in (30).

4.14.1 Expected Mean Squares and Variance Component Estimation

- For equal cell sizes (all $n_{ij} = n$):

$$\begin{aligned} E(MS_A) &= \sigma^2 \; + \; n\sigma^2_{\tau\beta} \; + \; \frac{bn}{a-1} \sum_{i=1}^{a} \tau_i^2 \\ E(MS_B) &= \sigma^2 \; + \; n\sigma^2_{\tau\beta} \; + \; an\sigma^2_\beta \\ E(MS_{AB}) &= \sigma^2 \; + \; n\sigma^2_{\tau\beta} \\ E(MS_E) &= \sigma^2 \end{aligned} \tag{34}$$

- These are the EMS for the **unrestricted mixed model**. This is the default in *SAS*.
- For unequal sample sizes, the coefficients of the variance components in the EMS is a complicated function of the n_i. *SAS* will generate them using the RANDOM statement.
- In the EMS output in SAS, replace 'Q' (for quadratic function of fixed effects) with $\dfrac{\sum(\text{fixed effect})^2}{df(\text{fixed effect})}$.
- Based on the expected mean squares, the appropriate test statistics are:

$$F_A \; = \; \frac{MS_A}{MS_{AB}} \text{ for (31),} \qquad F_B \; = \; \frac{MS_B}{MS_{AB}} \text{ for (32),} \qquad F_{AB} \; = \; \frac{MS_{AB}}{MS_E} \text{ for (33)}$$

- For the mixed effects model, replace $E(MS_B)$, $E(MS_{AB})$, and $E(MS_E)$ in (34) with the calculated values of MS_B, MS_{AB}, and MS_E. Solving the system of equations produces estimates of the variance components:
  - (Estimate of σ^2) $=$ $\widehat{\sigma}^2$ $= MS_E$.
  - (Estimate of $\sigma^2_{\tau\beta}$) $=$ $\widehat{\sigma}^2_{\tau\beta}$ $= \dfrac{MS_{AB} - MS_E}{n}$.
  - (Estimate of σ^2_{β}) $=$ $\widehat{\sigma}^2_{\beta}$ $= \dfrac{MS_B - MS_{AB}}{an}$.
- These estimates are generated using the PROC VARCOMP procedure. For any mixed effects model:
  - In the MODEL statement in PROC VARCOMP you must list all fixed effects before the random effects. Then include the **/ FIXED =** n option where n is the number of fixed effects.
  - In the following example, PRODRUN is the only fixed effect in the model. The model statement is **MODEL strength = prodrun batch prodrun*batch / FIXED = 1**.

4.14.2 Mixed Effects Model Example

During any workday there are only 4 production runs. Six batches of raw material were randomly selected from the large number of available batches. In each of the 4 runs, the same 6 batches of raw materials were used to make a synthetic fiber. Three random sample were taken from each (production run)*batch combination. The data is the same as the random effects model example except that now we treat production run as fixed.

TWO FACTOR MIXED EFFECTS MODEL

The GLM Procedure

Variable: strength

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	23	2295.503778	99.804512	385.75	<.0001
Error	48	12.418933	0.258728		
Corrected Total	71	2307.922711			

R-Square	Coeff Var	Root MSE	strength Mean
0.994619	2.053826	0.508653	24.76611

Source	DF	Type III SS	Mean Square	F Value	Pr > F
batch	5	1320.720844	264.144169	1020.93	<.0001
prodrun	3	920.650922	306.883641	1186.13	<.0001
prodrun*batch	15	54.132011	3.608801	13.95	<.0001

Source	Type III Expected Mean Square
batch	Var(Error) + 3 Var(prodrun*batch) + 12 Var(batch)
prodrun	Var(Error) + 3 Var(prodrun*batch) + Q(prodrun)
prodrun*batch	Var(Error) + 3 Var(prodrun*batch)

The GLM Procedure
Tests of Hypotheses for Mixed Model Analysis of Variance

Variable: strength

Source	DF	Type III SS	Mean Square	F Value	Pr > F
batch	5	1320.720844	264.144169	73.19	<.0001
prodrun	3	920.650922	306.883641	85.04	<.0001
Error	15	54.132011	3.608801		
Error: MS(prodrun*batch)					

Source	DF	Type III SS	Mean Square	F Value	Pr > F
prodrun*batch	15	54.132011	3.608801	13.95	<.0001
Error: MS(Error)	48	12.418933	0.258728		

TWO FACTOR MIXED EFFECTS MODEL
VARIANCE COMPONENTS ANALYSIS

Variance Components Estimation Procedure

Dependent Variable:	strength

REML Iterations				
Iteration	Objective	Var(batch)	Var(prodrun*batch)	Var(Error)
0	-17.7618754842	21.7112806790	1.1166909877	0.2587277778
1	-17.7618754842	21.7112806790	1.1166909877	0.2587277778

Convergence criteria met.

REML Estimates	
Variance Component	Estimate
Var(batch)	21.71128
Var(prodrun*batch)	1.11669
Var(Error)	0.25873